

Applications of Differential Forms in Thermodynamics

Frank Wang, fwang@lagcc.cuny.edu

We use Maple's **DifferentialGeometry** package to derive Maxwell and other mathematical identities in thermodynamics.

The laws of thermodynamics can be concisely stated: (I) the energy of the universe is constant; (II) the entropy of the universe is increasing. However, many students consider it as a difficult subject. In his classic textbook, Zemansky introduced some mnemonic schemes for students to recall the Maxwell identities. Feynman suggested that thermodynamics is hard because people use different variables: some choose temperature and volume, some choose temperature and pressure, and so on. Weyl pointed out that the inconsistent usage of symbols by mathematicians and physicists could cause confusion. This worksheet is designed to show the connections among equations so that students are not lost in the thermodynamic jungle of partial derivatives.

Consider a differentiable function of two variables, $z=f(x, y)$, the total differential shown in a mathematics textbook is the following.

$$dz = \frac{\partial z}{\partial x} dx + \frac{\partial z}{\partial y} dy$$

We can consider dz as a one form, and below is its representation using the **DifferentialGeometry** package.

```
> restart:
```

```
> with(DifferentialGeometry):
```

```
> DGsetup([x, y], M);
```

frame name: M (1)

```
M > ExteriorDerivative(z(x, y));
```

$$\left(\frac{\partial}{\partial x} z(x, y) \right) dx + \left(\frac{\partial}{\partial y} z(x, y) \right) dy \quad (2)$$

In thermodynamics, the notation is slightly different. Consider the internal energy U , if it is treated as a function of temperature T and volume V , physics or chemistry textbooks will write the total differential as

$$dU = \left(\frac{\partial U}{\partial T} \right)_V dT + \left(\frac{\partial U}{\partial V} \right)_T dV$$

However, U can be treated as a function of T and the pressure P , and total differential becomes

$$dU = \left(\frac{\partial U}{\partial T} \right)_P dT + \left(\frac{\partial U}{\partial P} \right)_T dP$$

The subscript is to remind us that U depends on another variable. We represent the two treatments below.

```
> restart:
```

```
> with(DifferentialGeometry):
```

```
> DGsetup([T, V], M);
```

frame name: M (3)

```
M > ExteriorDerivative(U(T, V));
```

$$\left(\frac{\partial}{\partial T} U(T, V) \right) dT + \left(\frac{\partial}{\partial V} U(T, V) \right) dV \quad (4)$$

```
M > DGsetup([T, P], N);
```

frame name: N (5)

N > ExteriorDerivative (U(T, P)) ;

$$\left(\frac{\partial}{\partial T} U(T, P) \right) dT + \left(\frac{\partial}{\partial P} U(T, P) \right) dP \quad (6)$$

From these examples, we understand that when we see in thermodynamics literature

$$\left(\frac{\partial U}{\partial T} \right)_P$$

it means $U = U(T, P)$, and when we see

$$\left(\frac{\partial U}{\partial V} \right)_T$$

it means $U = U(T, V)$, and so forth.

According to the first law of thermodynamics, the change of internal energy of a system is the sum of the change of heat and work done on a system. From the second law, there is a function, called entropy S , which is a total differential, or an exact one form in differential geometry language. Together we write

$$dU = T dS - P dV$$

This is the only physics condition we use.

For exact one forms such as dU and dS , their exterior derivative vanishes. This is known as Poincaré's lemma. With merely these two equations, essentially all the result of thermodynamics can be derived.

We first consider U as a function of the entropy S and volume V .

> restart:

> with(DifferentialGeometry): Preferences("ShowFramePrompt", false)
:

> DGsetup([S, V], M);

frame name: M (7)

> dU := evalDG(T(S, V)*dS - P(S, V)*dV)

$$dU := T(S, V) dS - P(S, V) dV \quad (8)$$

> ExteriorDerivative(dU);

$$-\left(\frac{\partial}{\partial S} P(S, V) + \frac{\partial}{\partial V} T(S, V) \right) dS \wedge dV \quad (9)$$

As stated earlier, the exterior derivative of dU is zero, which implies

$$\frac{\partial}{\partial S} P(S, V) + \frac{\partial}{\partial V} T(S, V) = 0$$

and in thermodynamics literature it is written as

$$\left(\frac{\partial P}{\partial S} \right)_V = - \left(\frac{\partial T}{\partial V} \right)_S$$

This is one of the Maxwell identities. If we choose S and P as the variables, we have the following.

> restart:

> with(DifferentialGeometry): Preferences("ShowFramePrompt", false)
:

> DGsetup([S, P], M);

frame name: M (10)

> dU := evalDG(T(S, P)*dS - P*ExteriorDerivative(V(S, P)));

$$dU := - \left(P \left(\frac{\partial}{\partial S} V(S, P) \right) - T(S, P) \right) dS - P \left(\frac{\partial}{\partial P} V(S, P) \right) dP \quad (11)$$

> ExteriorDerivative(dU);

$$-\left(- \frac{\partial}{\partial S} V(S, P) + \frac{\partial}{\partial P} T(S, P) \right) dS \wedge dP \quad (12)$$

The conclusion is

$$\left(\frac{\partial V}{\partial S}\right)_P = \left(\frac{\partial T}{\partial P}\right)_S$$

With T and V as independent variables,

> **restart:**

> **with(DifferentialGeometry): Preferences("ShowFramePrompt", false)**
:

> **DGsetup([T, V], M);**

frame name: M

(13)

> **dU := evalDG(T*ExteriorDerivative(S(T,V)) - P(T,V)*dV);**

$$dU := T \left(\frac{\partial}{\partial T} S(T, V) \right) dT + \left(-P(T, V) + T \left(\frac{\partial}{\partial V} S(T, V) \right) \right) dV$$

(14)

> **ExteriorDerivative(dU);**

$$- \left(\frac{\partial}{\partial T} P(T, V) - \frac{\partial}{\partial V} S(T, V) \right) dT \wedge dV$$

(15)

Here is another relation:

$$\left(\frac{\partial P}{\partial T}\right)_V = \left(\frac{\partial S}{\partial V}\right)_T$$

The exterior derivative of dS gives a basic equation:

> **dS := evalDG((ExteriorDerivative(U(T,V)) + P(T,V)*dV)/T);**

$$dS := \frac{\left(\frac{\partial}{\partial T} U(T, V) \right) dT}{T} + \frac{\left(P(T, V) + \frac{\partial}{\partial V} U(T, V) \right) dV}{T}$$

(16)

> **ExteriorDerivative(dS);**

$$- \frac{\left(-T \left(\frac{\partial}{\partial T} P(T, V) \right) + P(T, V) + \frac{\partial}{\partial V} U(T, V) \right) dT}{T^2} \wedge dV$$

(17)

The first energy equation is

$$\left(\frac{\partial U}{\partial V}\right)_T = T \left(\frac{\partial P}{\partial T}\right)_V - P$$

Chemists prefer to use T and P as variables.

> **restart:**

> **with(DifferentialGeometry): Preferences("ShowFramePrompt", false)**
:

> **DGsetup([T, P], M);**

frame name: M

(18)

> **dU := evalDG(T*ExteriorDerivative(S(T,P)) - P*ExteriorDerivative(V(T,P)));**

$$dU := - \left(P \left(\frac{\partial}{\partial T} V(T, P) \right) - T \left(\frac{\partial}{\partial T} S(T, P) \right) \right) dT - \left(P \left(\frac{\partial}{\partial P} V(T, P) \right) - T \left(\frac{\partial}{\partial P} S(T, P) \right) \right) dP$$

(19)

> **ExteriorDerivative(dU);**

$$\left(\frac{\partial}{\partial P} S(T, P) + \frac{\partial}{\partial T} V(T, P) \right) dT \wedge dP$$

(20)

The last Maxwell identity is

$$\left(\frac{\partial S}{\partial P}\right)_T = - \left(\frac{\partial V}{\partial T}\right)_P$$

By writing dS as a function of T and P

> `dS := evalDG((ExteriorDerivative(U(T, P)) + P*ExteriorDerivative(V(T, P)))/T);`

$$dS := \frac{\left(P \left(\frac{\partial}{\partial T} V(T, P)\right) + \frac{\partial}{\partial T} U(T, P)\right) dT}{T} + \frac{\left(P \left(\frac{\partial}{\partial P} V(T, P)\right) + \frac{\partial}{\partial P} U(T, P)\right) dP}{T} \quad (21)$$

> `ExteriorDerivative(dS);`

$$-\frac{\left(P \left(\frac{\partial}{\partial P} V(T, P)\right) + \left(\frac{\partial}{\partial T} V(T, P)\right) T + \frac{\partial}{\partial P} U(T, P)\right) dT}{T^2} \wedge dP \quad (22)$$

we obtain the second energy equation

$$\left(\frac{\partial U}{\partial P}\right)_T = -T \left(\frac{\partial V}{\partial T}\right)_P - P \left(\frac{\partial V}{\partial P}\right)_T$$

Enthalpy H is defined as the following.

$$dH = dU + d(PV) = T dS + V dP$$

> `dS1 := evalDG((ExteriorDerivative(H(T, P)) - V(T, P)*dP)/T);`

$$dS1 := \frac{\left(\frac{\partial}{\partial T} H(T, P)\right) dT}{T} + \frac{\left(-V(T, P) + \frac{\partial}{\partial P} H(T, P)\right) dP}{T} \quad (23)$$

> `ExteriorDerivative(dS1);`

$$-\frac{\left(\left(\frac{\partial}{\partial T} V(T, P)\right) T + \frac{\partial}{\partial P} H(T, P) - V(T, P)\right) dT}{T^2} \wedge dP \quad (24)$$

The second energy equation is equivalent to

$$\left(\frac{\partial H}{\partial P}\right)_T = -T \left(\frac{\partial V}{\partial T}\right)_P + V$$

Appendix

In this appendix we illustrate that the change of heat dQ is not exact, but is integrable with a proper integrating factor.

We express the first law for an ideal gas as

$$dQ = dU + P dV = C_V dT + P dV = C_V dT + \frac{RT}{V} dV$$

The first energy equation indicates that $U(T, V)$ is independent of V for an ideal gas, which has an equation of state $P \cdot V = R \cdot T$, where R is the gas constant. The partial derivative $U(T, V)$ with respect to T is defined as the capacity C subscript V . Its value for an ideal monoatomic gas is $\frac{3}{2} \cdot R$. We show that dQ is not exact because its exterior derivative is not zero.

> `restart;`
 > `with(DifferentialGeometry): Preferences("ShowFramePrompt", false);`
 > `DGsetup([T, V], M);`

$$\text{frame name: } M \quad (25)$$

> `dQ := evalDG(C[V]*dT + R*T/V*dV);`

$$dQ := C_V dT + \frac{R T dV}{V} \quad (26)$$

> `ExteriorDerivative(dQ);`

$$\frac{R dT}{V} \wedge dV \quad (27)$$

If we multiply dQ by an integrating factor $\frac{1}{T}$, called dS , then it becomes exact.

> `dS = evalDG(dQ/T);`

$$dS = \frac{C_V dT}{T} + \frac{R dV}{V} \quad (28)$$

> `ExteriorDerivative(dS);`

$$0 dT \quad (29)$$

We can solve this differential equation.

> `soln := dsolve(diff(T(V),V) = -R/V/(C[V]/T(V)), T(V));`

$$\text{soln} := T(V) = _C1 V^{-\frac{R}{C_V}} \quad (30)$$

Notice that the solution is the adiabatic curve, which satisfies the following condition:

$$TV^{\gamma-1} = \text{constant where } \gamma = \frac{5}{3} \text{ for monoatomic gases}$$

We use a vector field to represent the one form.


> `with(plots):`

> `R := 1; C[V] := 3/2*R;`

$$\begin{aligned} R &:= 1 \\ C_V &:= \frac{3}{2} \end{aligned} \quad (31)$$


For dQ ,

> `fieldplot([R*T/V, C[V]], V=1..3, T=1..3, arrows = slim);`



For dS ,

> `pf := fieldplot([R/V, C[V]/T], V=1..3, T=1..3, arrows = slim);`



We plot several adiabatic curves.


```
> adl := seq(rhs(soln)/2, _C1=2..12);
```

$$adl := \frac{1}{v^{2/3}}, \frac{3}{2v^{2/3}}, \frac{2}{v^{2/3}}, \frac{5}{2v^{2/3}}, \frac{3}{v^{2/3}}, \frac{7}{2v^{2/3}}, \frac{4}{v^{2/3}}, \frac{9}{2v^{2/3}}, \frac{5}{v^{2/3}},$$

$$\frac{11}{2v^{2/3}}, \frac{6}{v^{2/3}}$$


(32)

```
> pad := plot({adl}, v=1..3, 1..3);
```



The figure below illustrates that the entropy one form is the gradient of the adiabatic curves.

```
> display({pf, pad});
```



References

1. The Feynman Lectures on Physics, Chapters 44 and 45, available at <https://www.feynmanlectures.caltech.edu>
2. Mark W. Zemansky, *Heat and Thermodynamics*, fifth edition. P. 285; pp. 291-292.
3. Hermann Weyl, *Space, Time, Matter*. Pp. 98-99.
4. Bernard Schutz, *Geometrical methods of mathematical physics*. Pp. 163-165.